PharmaSUG 2015 – Paper PO06

Programming Pharmacokinetic (PK) Timing and Dosing Variables in Oncology Studies: Demystified

Kiran Cherukuri, Seattle Genetics Inc., Bothell, WA

ABSTRACT

Pharmacokinetic (PK) analysis is a major part of clinical trials intended to characterize the time course of the drug and/or metabolite concentrations after drug administration in order to obtain information on drug disposition in humans.

This paper discusses about how timing and dosing variables are derived through a generic program to facilitate PK analysis. In oncology studies, the last timepoint of one cycle is often also the first timepoint of the subsequent cycles. Accommodating multiple timepoints for a single PK concentration data point on two different time scales is important to facilitate analysis, but presents a programming challenge. This challenge is described and solutions provided. In addition, the significance of these variables as they are used for concentration time profiles and derivation of PK parameters such as AUC, Cmax, and Tmax is presented.

INTRODUCTION

The use of Clinical Data Interchange Standards Consortium (CDISC) standards for submission of clinical trial data is widely practiced and implemented in our industry. Submission of Phamacokinetic (PK) data uses the same standards; Study Data Tabulation Model (SDTM) Implementation Guide defines the standards for Pharmacokinetic Concentration (PC) data and Pharmacokinetic Parameter (PP) data. Analysis Data Model (AdaM) defines the standards for Pharmacokinetic Concentration (ADPC) data and Pharmacokinetic Parameters (ADPP) data. These SDTM/ADaM datasets are generated based on metadata, clinical data, bioanalytical data and the derived PK parameters calculated by pharmacokineticists. All of the PK related tables, figures, and listings can be generated based on analysis datasets, ADPC and ADPP.

The general workflow for the analysis of PK data in clinical trials involves two major steps,

1) Generating a PK Concentration file
2) Calculation of PK parameters

This paper discusses standardizing the generation of PK concentration files by clinical programmers in compliance with CDISC standards which is used by clinical pharmacologist to generate PK parameters. The PK parameter file can then be used to generate the associated SDTM and ADaM files.

PROCESS FLOW

The following diagram presents one option for transferring research sample data from the BioAnalytical group through Biometrics to the Clinical Pharmacology team for PK analysis and back to the Biometrics group for reporting.



FUNCTIONAL DESCRIPTION

The program reads data from SDTM PC, treatment variables from DM and date/ dosing information from EX datasets to generate a customized PK concentration file for the clinical pharmacology group to perform PK analysis. This file includes nominal and actual dose and time variables, concentration variables and a few other variables which are not generally collected in the SDTM PC dataset. Below are key functions performed by the program.

1. Planned time point variable (PCTPT) values, planned time point number variable (PCTPTNUM) values and calculated cycle day variable values are entered as input strings. The calculations to obtain the values as strings are done within the program.
2. Dosing information is pulled from EX dataset. Initial dosing information and dosing at each cycle is collected separately.
3. Visit cycle values are derived from visit variable and PK analysis cycle is derived based on visit cycle and visit number.
4. Predose records of each cycle, excluding cycle 1, are repeated to represent a single concentration on two different time scales; example, for PK sample collection on various timepoints within a cycle, Day 1 (predose), 2, 3, 8 and 15, predose of the subsequent cycle (ex: cycle 2) is repeated and the time point is collected as day 21 of the current cycle (ex: cycle1 day 21). In PK analysis without the addition of the new record, predose of the subsequent cycle will not be included in the PK analysis.

   The code below derives PK analysis cycle values and creates multiple records for pre-dose draws.

```
data win0;
  set s.pc(where=(pctptnum > .z and pctestcd ^in ('ATA')));
  if upcase(visit)^=:"UNSCHEDULED" and upcase(visit)^=:"END OF TREATMENT" then
cycle=input(scan(visit,2,''),cycle.);
  else cycle=input(visit,cycle.);
  l_cycle=lag(cycle);
run;

proc sort data=win0;
  by usubjid pctest cycle pcdtc visit;
run;

*-------------------------------------------------------------------------*
| Calculate Winolin Cycle; handle multiple doses within a cycle
*-------------------------------------------------------------------------*;
data win0(drop=l_cycle);
  set win0;
  by usubjid pctest cycle pcdtc visit;
  if cycle ^ in (66 77) then do;
      wnlcycle=cycle;
    output;
  end;
  else do;
      wnlcycle=l_cycle;
    output;
  end;

  if first.cycle and cycle ^in(. 1 66 77) and
(input(substr(scan(visit,4,''),1),best.)=1 or
  index(upcase(pctpt),'PREDOSE')>0) then do;
      wnlcycle=cycle-1;
      wnl=wnlcycle;
      output;
  end;
run;

proc sort data=win0;
  by usubjid pctest visitnum pctptnum cycle;
run;
data win0;
  set win0;
  by usubjid pctest visitnum pctptnum cycle;

  retain pctptn cyclen flag;
  if first.pctest then do; pctptn=pctptnum; cyclen = cycle; flag = ' '; end;
```

```
  else do;
       if pctptnum > pctptn then pctptn = pctptnum;
       else do;
           flag = 'Y';
              if cycle = cyclen then pctptn = pctptn;
           else pctptn = pctptnum;
       end;
  end;
run;
proc sort data=win0;
  by usubjid pctest visitnum pctptn;
run;
data win0;
  set win0;
  by usubjid pctest visitnum pctptn;
  retain cycle_n;
  if first.pctptn and flag ne 'Y' then cycle_n = cycle;
  else if wnl eq . and cycle eq cyclen then cycle_n = cycle + 0.1;
  else cycle_n = cycle;
run;
proc sort data=win0(drop=flag cyclen pctptn);
  by usubjid pctest cycle pcdtc visit;
run;

*------------------------------------------------------------------------*
| Get SDTM EX and DM dataset; Get Dosing information, SDTM EX dataset. Generate
| two datasets, one to get initial dosing information (SGDOSE) and one to get
| dosing at every time point (EX)
*------------------------------------------------------------------------*;
proc sort data=s.ex out=ex;
  by usubjid;
run;
proc sort data=s.dm out=dm(keep=usubjid armcd);
  by usubjid;
run;

*------------------------------------------------------------------------*
| Merge EX and DM dataset; handle multiple doses within a cycle for
| a proper merge
*------------------------------------------------------------------------*;
data ex;
  merge ex(in=a) dm;
  length cohort $20;
  by usubjid;
  if a;
  cohort=compbl('ARM '||armcd);
run;

proc sql;
 create table sgdose as
 select distinct usubjid, visit, cohort, exdosrgm, exstdtc as sgdtc, exendtc as
endtc
 from ex
 where exdose >.z
 group by usubjid
 having exstdtc = min(exstdtc)
 order by usubjid;
quit;

proc sort data=ex out=ex2(keep=usubjid visit exdosrgm exdose exstdtc exendtc
exstdy rename=(exdosrgm=dose_gvn)) nodupkey dupout=dupex;
  by usubjid visit exstdtc;
  where exstdtc ne ' ';   /* Missing exposure dates should be excluded */
run;
data ex2;
  set ex2;
  by usubjid visit exstdtc;
```

```
  if upcase(visit)^=:"UNSCHEDULED" and upcase(visit)^=:"END OF TREATMENT" then
cycle=input(scan(visit,2,''),cycle.);
  else cycle=input(visit,cycle.);
run;

proc sort data=ex2;
  by usubjid cycle visit exstdtc ;
run;

data ex2;
  set ex2;
  by usubjid cycle visit exstdtc;

  retain cycle_n;
  if first.cycle then cycle_n = cycle;
  else cycle_n = cycle_n + 0.1;
run;
proc sort data=ex2;
  by usubjid exstdtc cycle;
run;

*-------------------------------------------------------------------------*
| Get previous dose/cycle exposure dates for creation of multiple records
*-------------------------------------------------------------------------*;
data ex3;
  set ex2;
  by usubjid exstdtc cycle;
  l_exsdtc=lag(exstdtc);
  if  scan(visit,2,'')  in  ('1'  '1A')  and  scan(visit,4,'')  eq  '1'  then  do;
l_exsdtc=''; end;
run;

proc sort data=win0;
  by usubjid cycle cycle_n;
run;

proc sort data=ex3(drop=visit);
  by usubjid cycle cycle_n;
run;

*-------------------------------------------------------------------------*
| Merge EX and WIN0 datasets to get dosing information at every timepoint
| and Calculate Day values
*-------------------------------------------------------------------------*;
data win1;
  merge win0 (in=a) ex3;
  by usubjid cycle;
  if a;
  if  index(lowcase(visit),’cycle’) then day =
  input(substr(scan(visit,4,''),1),best.);
  else if lowcase(visit) eq 'end of treatment' then day = 77;
  else if index(lowcase(visit), 'unscheduled') then day = 66;

  if index(pctpt, 'PREDOSE') then do;
     if day ne 1 then put 'WAR' 'NING: pctpt is PREDOSE but day ne 1 ' usubjid=
visit= pcrefid=;
  end;
run;

data win1;
  set win1;
  by usubjid cycle;
  if wnl ne . then do; exstdtc=l_exsdtc; end;
run;

proc sort data = win1(drop=l_exsdtc);
  by usubjid;
```

```
run;
proc sort data = sgdose;
  by usubjid;
run;

*------------------------------------------------------------------------*
| Merge WIN and SGDOSE datasets to get intital dosing information
*------------------------------------------------------------------------*;
data win2;
  merge win1(in = a) sgdose(in = b);
  by usubjid;
  if a * ~b then put 'WAR' 'NING: Subject in PC but not in EX: ' usubjid=;
  if b * ~a then put 'NOT' 'CIE: Subject dosed but not in PC: ' usubjid=;
  if a;
run;
proc sort data=win2;
  by usubjid pctest wnlcycle cycle exstdtc pctptnum;
run;
```

5. A dummy dataset is created based on the PK concentration specifications. This is done to include all the variables in the output file irrespective of any missing data.

```
*------------------------------------------------------------------------*
| Import a dummy dataset and merge it to the final pcc dataset
*------------------------------------------------------------------------*;
libname xls odbc noprompt="Driver={Microsoft Excel Driver (*.xls, *.xlsx,
*.xlsm, *.xlsb)};
 DBQ=O:\Admin\Programming\Infrastructure\PK
Standards\PK_Concentration_Specs_01Oct2013_Updated.xls";
data _null_;
       length var $32 fmt $32 varlist labl $32000;
       set xls.'pk_specs$'n;
       retain varlist labl;
       if ^missing(input(File_Conventions, ?? 8.)) then do;
             x+1;
             var=prxchange("s/[^A-Z0-9_]//i",-1,strip(f2));
             varlbl=prxchange("s/[^A-Z0-9_]//i",-1,strip(f3));
             varlist=catx(", ",strip(varlist),strip(var));

       fmt=ifc(strip(f4)=:'C',"$"||strip(put(strip(f5),8.)),strip(put(strip(f5),
8.)));
             labl=catx("                              ",strip(labl),strip(var)||"
label="||"'"||strip(varlbl)||"'";
             call              symputx('varfmt'||strip(put(x,8.)),strip(var)||"
length="||strip(fmt),'G');
             call symputx('var'||strip(put(x,8.)),strip(var),'G');
             call symputx('varlist',strip(varlist),'G');
             call symputx('nvar',x,'G');
             call symputx('labl',strip(labl),'G');
       end;
run;
libname xls clear;
%macro dummy();
data dummy;
  %do i=1 %to &nvar;
       attrib &&varfmt&i;
    call missing(&&var&i);
  %end;
run;
%mend;
%dummy();
```

6. Finally, once the file is ready with all the required variables, an export procedure is performed to import the data to excel.

Input Variables

Apart from SDTM PC, DM and EX dataset variables, following input strings are used to generate the PK concentration file.

PCNOM: The string of planned time point (PCTPT) values in minutes. Usually this is collected in hours (ex: “predose of xxxx”, “0min”, “24h”, etc.). Convert all time points into minutes and input all discrete values to a string. Do not include EOT time points. For predose this would be 0 and for '0min' or 'End of study drug administration' the value would be what the study team decides, it could be 1 min or 30 min or 60min.

```
%let pcnom = 0 1 1440 2880 10080 20160;
```

PCNUM: Planned time point number (PCTPTNUM). All discrete values of PCTPTNUM in the dataset are input to this variable as a string.

VIS: The cycle day variable which is calculated in the program. Values of cycle day variable are entered as a string into this variable.

```
*------------------------------------------------------------------------*
| Macro to get PCNUM(PCTPTNUM values) and VIS(Cycle day value from VISIT)
| strings used in deriving Nominal time point variables
*------------------------------------------------------------------------*;
%macro getval(inds=, var=, subset=);
%global outvar;
  %let dsid=%sysfunc(open(&inds));
  %if %sysfunc(varnum(&dsid,&var)) >0 %then %do;
    proc sql noprint;
         select distinct(&var) into: outvar separated by ' '
         from &inds
         &subset;
    quit;
  %end;
  %let cl=%sysfunc(close(&dsid));
%mend;

*------------------------------------------------------------------------*
| VIS, using %getval() to generate VIS string
*------------------------------------------------------------------------*;
%getval(inds=win3, var=cday, subset=%str(where cycle not in (66 77)));

*------------------------------------------------------------------------*
| PCNUM, using %getval() to generate PCNUM string
*------------------------------------------------------------------------*;
%getval(inds=win3, var=pctptnum, subset=%str(where upcase(pctpt) ne "EOT" and
upcase(pctpt) ne "END OF TREATMENT" and upcase(pctpt) ne "NON-DOSING"));
```

Output Variables

Additional observation, for predose records, **representing a single concentration on two different time scales**

DUPLCATE: Flag to indicate duplicate created record. For Pre-dose samples of each dose (not just each cycle) except C1D1, sample should be entered twice in file - once as last sample from previous dose, and once as predose to current dose. The new record (last sample from previous dose) should be marked as Duplicate. If no Pre-dose sample record exists, duplicate record should not be created. Time values for end-of-dose records should be based on previous dose or cycle (as appropriate).

PK Analysis Cycle

WNLCYCLE: Same as study cycle except for duplicate created records, for which this value is cycle – 1, cycle is the value of the visit cycle. For Unscheduled and EOT, WNLCYCLE should be adjusted to be the same as the previous sample in time

Nominal and Actual time point and dose

Nominal is defined as when something is planned, scheduled or expected to happen. Actual refers to when it actually does happen. Thus, Nominal Dose refers to how many doses of drug the patient should have had up to that point, whereas Actual Dose denotes how many doses the patient has actually received. Similarly, Nominal Time is the planned time for the sample based on the Nominal time point, and the Actual Time is when the sample was actually drawn from the patient based on the entered date and time.

ACTDOSE: Actual dose number across all cycles, may differ from the nominal dose if a dose was skipped

NOMDOSE: The nominal dose across all cycles, based on number of doses the patient should have received

NOMTM: Nominal time point, from CRF

NOMCUMTM: Nominal time from first dose (day) of the study (cycle 1 day 1). This is set to missing for Unscheduled, EOT and LTFU samples.

NOMCYCTM: Nominal time from first dose (day) of this WNLCYCLE. This is set to missing for Unscheduled, EOT and LTFU samples. For the duplicate record, calculate based on the assigned cycle.

NOMDSTM: Nominal time from previous dose (day) within this WNLCYCLE. This is set to missing for Unscheduled, EOT and LTFU samples. For the duplicated record, calculate based on the previous dose within the assigned cycle. This value is equal to NONCYCTM, except when multiple doses are given per cycle.

ACTCUMTM: Actual time from first dose (day) of the study (cycle 1 day 1). Pre-dose of study is manually set to 0.

ACTCYCTM: Actual time from first dose (day) of this WNLCYCLE. Pre-dose of cycle manually set to 0. For the duplicated record, calculate based on the assigned cycle.

ACTDSTM: Actual time from previous dose (day) within this WNLCYCLE. Pre-dose of dose manually set to 0; round to 4 decimal places. For the duplicated record, calculate based on the previous dose within the assigned cycle. This value is equal to ACTCYCTM, except when multiple doses are given per cycle.

Other Variables derived

WNL_NGML: Result in ng/ml except when concentration is less than LLQ. When result is less than LLQ set pre-dose values to zero and post-dose to missing

WNL_UGML: Result in ug/ml except when concentration is less than LLQ. When result is less than LLQ set pre-dose values to zero and post-dose to missing

ORGCONC: original result from BA lab. BLOQ should be noted with BLOQ<(XXX)

ORGUNITS: The units (ex: ng/mL or ug/mL) that are associated with the original result from the BA lab.

DIFCUMTM: Difference of ACTCUMTM (Actual Cumulative Time) and NOMCUMTM (Nominal Cumulative Time)

DIFCYCTM: Difference of ACTCYCTM (Actual Cycle Time) and NOMCYCTM (Nominal Cycle Time)

DIFDSTM: Difference of ACTDSTM (Actual Dose Time) and NOMDSTM (Nominal Dose Time)

SIGNIFICANCE OF DERIVED VARIABLES

Mean Concentration Time Profiles

Concentration time profiles use variables provided in the PK concentration file. The graphical example in Figure 1, establishes the significance of variables NOMCUMTM and WNL_NGML. Other variables, like NOMCYCTM, are used if a time profile has to be generated by cycle.



Figure1. The Mean Concentration Time Profile

Pharmacokinetic parameters

PK parameters, AUC, Cmax, Ctrough, etc. generated depend on the variables WNL_NGML or WNL_UGML present in the PK concentration file. A screen shot of the parameters generated is shown below.

- Area under the curve ($AUC_{0\text{-}21d}$) and $AUC_{0\text{-}\infty}$
- Concentration at the end of infusion (C_{eoi}) or maximum observed concentration (C_{max})
- Trough concentration (C_{trough})
- Terminal or apparent terminal half-life ($t_{1/2}$)
- Systemic clearance and volume of distribution at steady state

File Window

COHORT	DOSE_GROUP	SUBJECT	AGEGR	HL_Lambda_z (day)	AUCINF_obs (day*ug/mL)	AUC21 (day*ug/mL)	Ceoi (ug/mL)	Ctrough (ug/mL)	Cl_obs (L/day)
ARM 1	xxx	xxx - xxxx	1.00	6.46	65.49	61.58	36.14	0.43	1.83
			N	**1**	**1**	**1**	**1**	**1**	**1**
			Mean	**6.459**	**65.491**	**61.585**	**36.144**	**0.425**	**1.832**
			SD						
			Median	**6.46**	**65.49**	**61.58**	**36.14**	**0.43**	**1.83**
			Geometric Mean	**6.459**	**65.491**	**61.585**	**36.144**	**0.425**	**1.832**
			CV% Geometric Mean						
ARM 1	xxx	xxx - xxxx	1.00	4.04	119.89	117.49	57.94	0.44	1.36
			2.00	5.49	100.37	93.66	32.31	0.86	1.04
			2.00	4.09	133.00	128.82	62.12	0.72	1.02
			2.00				34.23	1.32	
			1.00	5.95	98.70	93.08	38.20	0.66	1.24
			2.00	4.31	83.67	80.52	38.28	0.33	1.27
			1.00	3.18	89.94	89.11	34.73	0.23	2.33
			2.00	4.75	81.99	79.02	31.06	0.49	0.96
			2.00	6.48	133.67	124.56	54.61	0.98	0.96
			2.00				43.61	1.03	
			1.00	6.90	102.24	95.13	35.50	0.71	1.57
			2.00	8.80	101.25	89.19	37.31	0.88	0.84
			2.00	5.19	122.46	117.53	60.64	0.50	1.39
			2.00					0.71	
			1.00					0.70	
			1.00	6.23	82.68	77.59	26.82	0.56	1.84
			1.00	7.05	137.42	126.53	52.08	1.07	1.31
			2.00	6.73	149.65	138.09	64.32	0.74	1.12
			1.00				55.34	0.98	
			2.00					0.53	
			2.00	9.13	120.73	105.66	64.82	1.15	1.20

CONCLUSION

This program is designed to simplify and standardize the process of generating PK concentration data for clinical pharmacology to perform PK analysis and to reduce programming and clinical pharmacology effort in producing and analyzing data. Planned time points, planned time point number and cycle values are critical variables for this program. For studies where any or all of these variables are not collected, appropriate variables should be used to calculate Nominal dose and time, PK analysis cycle values.

ACKNOWLEDGEMENTS

I would like to thank Rajeev Karanam, Assoc. Director, Clinical Programming, for encouraging me to write this paper, my Manager Jay Gadhiya for reviewing the paper and Jing Yang, Clinical Pharmacologist and Liz Thomas, Sr. Biostatistician for assisting with this project and reviewing the paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Kiran Cherukuri
Seattle Genetics, Inc.
21823 - 30th Drive S.E.
Bothell, WA 98021
425-527-4868
kcherukuri@seagen.com